

# Combinational Logic Design: Part 6: 8-bit Hierarchical Comparator

*Author Adil R.*

*May 2025*

## 1 Introduction

This report presents the design and implementation of an 8-bit hierarchical comparator using Verilog HDL. The comparator extends the basic 4-bit design to handle larger numbers while maintaining efficient logic utilization. The circuit compares two 8-bit binary numbers and produces three outputs indicating their relationship (greater than, equal to, or less than).

## 2 Design Description

The 8-bit comparator is implemented using a hierarchical approach with two 4-bit comparator modules and additional logic for combining results.

### 2.1 Architecture

- Two 4-bit comparator modules compare the upper and lower nibbles
- Combination logic resolves final output based on nibble comparisons
- Three output signals: G (Greater), E (Equal), L (Less)

### 2.2 Truth Table

Upper Nibble (A[7:4], B[7:4])		Lower Nibble (A[3:0], B[3:0])		Output		
G	E	G	E	G	E	L
1	0	X	X	1	0	0
0	1	1	0	1	0	0
0	1	0	1	0	1	0
0	1	0	0	0	0	1
0	0	X	X	0	0	1

Table 1: 8-bit Comparator Truth Table (X = Don't Care)

### 2.3 Boolean Equations

The output logic can be expressed as:

$$G = G_{upper} + (E_{upper} \cdot G_{lower})$$

$$E = E_{upper} \cdot E_{lower}$$

$$L = L_{upper} + (E_{upper} \cdot L_{lower})$$

## 3 Verilog Implementation

### 3.1 4-bit Comparator Module

```
1  `timescale 1ns / 1ps
2
3  module comp_4bit(
4      input  [3:0] A, B,
5      output reg G, E, L
6  );
7
8  always @(*) begin
9      G = 1'b0;
10     E = 1'b0;
11     L = 1'b0;
12
13     if (A > B)
14         G = 1'b1;
15     else if (A == B)
16         E = 1'b1;
17     else
18         L = 1'b1;
19 end
20
21 endmodule
```

### 3.2 8-bit Hierarchical Comparator

```
1  `timescale 1ns / 1ps
2
3  module comp_8bit(
4      input  [7:0] A, B,
5      output G, E, L
6  );
7
8  wire G_upper, E_upper, L_upper;
9  wire G_lower, E_lower, L_lower;
10
11 // Upper nibble comparison
12 comp_4bit upper_comp(
13     .A(A[7:4]),
14     .B(B[7:4]),
15     .G(G_upper),
16     .E(E_upper),
17     .L(L_upper)
18 );
19
20 // Lower nibble comparison
21 comp_4bit lower_comp(
22     .A(A[3:0]),
23     .B(B[3:0]),
24     .G(G_lower),
25     .E(E_lower),
26     .L(L_lower)
27 );
28
29 // Output logic
30 assign G = G_upper | (E_upper & G_lower);
31 assign E = E_upper & E_lower;
32 assign L = L_upper | (E_upper & L_lower);
33
34 endmodule
```

## 4 Test Bench

```
1  `timescale 1ns / 1ps
2
3  module tb_comp_8bit;
4      reg [7:0] A, B;
5      wire G, E, L;
6
7      comp_8bit uut(
8          .A(A),
9          .B(B),
10         .G(G),
11         .E(E),
12         .L(L)
13     );
14
15     initial begin
16         // Test case groups
17         // 1. Equal numbers
18         A = 8'h00; B = 8'h00;
19         #10 A = 8'hFF; B = 8'hFF;
20         #10 A = 8'hA5; B = 8'hA5;
21
22         // 2. A > B cases
23         #10 A = 8'h01; B = 8'h00; // Lower difference
24         #10 A = 8'h10; B = 8'h0F; // Upper difference
25         #10 A = 8'hF1; B = 8'hF0; // Both different
26
27         // 3. A < B cases
28         #10 A = 8'h00; B = 8'h01;
29         #10 A = 8'h0F; B = 8'h10;
30         #10 A = 8'hF0; B = 8'hF1;
31
32         // 4. Boundary cases
33         #10 A = 8'h80; B = 8'h7F; // MSB difference
34         #10 A = 8'h7F; B = 8'h80;
35         #10 A = 8'hFF; B = 8'h00;
36         #10 A = 8'h00; B = 8'hFF;
37
38         #20 $finish;
39     end
40
41     initial begin
42         $monitor("Time = %t, A = %h, B = %h, G = %b, E = %b, L = %b",
43             $time, A, B, G, E, L);
44     end
45 endmodule
```

## 5 Conclusion

- Successfully designed and implemented an 8-bit hierarchical comparator
- Testbench verified all comparison cases




Figure 1: Hierarchical 8-bit Comparator Schematic




Figure 2: Simulation Waveform Showing All Comparison Cases